New development in freefem++
Frédéric Hecht

To cite this version:
New Developments in FreeFem++

F. HECHT

Received May 30, 2012 Communicated by Yu. A. Kuznetsov
Received in revised form March 1, 2012

Abstract — This is a short presentation to the capability of the freefem++ software, in section 1, we recall most of the characteristics of the software. In section 2, we recall how to to build a weak form form of an partial differential equation (PDE) from the strong form of the PDE. In three last sections, we present different problem, tools to illustrated the software. First we do mesh adaptation problem in two and three dimension., secondly, we solve numerically a Phase change with Natural Convection, and the finally to show the HPC possibility we show a Schwarz Domain Decomposition problem on parallel computer.

Keywords: Finite Element

1. Introduction

This paper intends to give a small presentation of the software FreeFem++. A partial differential equation is a relation between a function of several variables and its (partial) derivatives. Many problems in physics, engineering, mathematics and even banking are modeled by one or several partial differential equations.

FreeFem++ is a software to solve these equations numerically in dimensions two or three. As its name implies, it is a free software based on the Finite Element Method; it is not a package, it is an integrated product with its own high level programming language; it runs on most UNIX, WINDOWS and MacOs computers. Moreover FreeFem++ is highly adaptive. Many phenomena involve several coupled systems, for example: fluid-structure interactions, Lorentz forces for aluminum casting and ocean-atmosphere coupling. Multiphysics problems require different finite element approximations or different polynomial degrees, possibly on different meshes. Some algorithms like Schwarz’ domain decomposition method also require data interpolation on multiple meshes within one program. FreeFem++ can handle these difficulties, i.e. arbitrary finite element spaces on arbitrary unstructured and adapted meshes.

*UPMC, Univ. Paris 06, UMR 7598, F-75005 Paris, France. E-mail: frederic.hecht@upmc.fr
This work was partially supported by ANR grant No. 01-2345-6789.
The characteristics of FreeFem++ are:

- Problem description (real or complex valued) by their variational formulations, with access to the internal vectors and matrices if needed.

- Multi-variables, multi-equations, bi-dimensional and three-dimensional static or time dependent, linear or nonlinear coupled systems; however the user is required to describe the iterative procedures which reduce the problem to a set of linear problems.

- Easy geometric input by analytic description of boundaries by pieces; however this part is not a CAD system; for instance when two boundaries intersect, the user must specify the intersection points.

- Automatic mesh generator, based on the Delaunay-Voronoï algorithm; the inner point density is proportional to the density of points on the boundaries [17].

- Metric-based anisotropic mesh adaptation. The metric can be computed automatically from the Hessian of any FreeFem++ function [20].

- High level user friendly typed input language with an algebra of analytic or finite element functions.

- Multiple finite element meshes within one application with automatic interpolation of data on different meshes and possible storage of the interpolation matrices.

- A large variety of triangular finite elements: linear, quadratic Lagrangian elements and more, discontinuous P1 and Raviart-Thomas elements, elements of a non-scalar type, the mini-element,... but no quadrangles.

- Tools to define discontinuous Galerkin finite element formulations $P_0, P_{1dc}, P_{2dc}$ and keywords: $\text{jump}, \text{mean}, \int_{\text{alledges}}$.

- A large variety of linear direct and iterative solvers (LU, Cholesky, Crout, CG, GMRES, UMFPACK [13], MUMPS [2], SuperLU, ...) and eigenvalue and eigenvector solvers (ARPARK) [24] and Optimization tools like Ipopt [33].

- Near optimal execution speed (compared with compiled C++ implementations programmed directly).

- Online graphics, generation of .txt, .eps, .gnu, mesh files for further manipulations of input and output data.

- Many examples and tutorials: elliptic, parabolic and hyperbolic problems, Navier-Stokes flows, elasticity, Fluid structure interactions, Schwarz’s domain decomposition method, eigenvalue problem, residual error indicator, ...
A parallel version using mpi

2. Weak formulation

For the first example consider a Poisson problem on a Domain $\Omega$ with a partition of the boundary $\partial \Omega$ in $\Gamma_2, \Gamma_e$: Find $u$ such that:

$$-\Delta u = 1 \text{ in } \Omega, \quad u = 2 \text{ on } \Gamma_2, \quad \frac{\partial u}{\partial n} = 0 \text{ on } \Gamma_e$$  \hspace{1cm} (2.1)

Denote $V_g = \{v \in H^1(\Omega) \mid v|_{\Gamma_2} = g\}$, and the weak form of the equation is obtained by multiplying by $v$ and integrating by parts (2.1). The variational formulation is: find $u \in V_2(\Omega)$, such that

$$\forall v \in V_0(\Omega), \quad \int_\Omega \nabla u \cdot \nabla v = \int_\Omega 1v + \int_{\Gamma} \frac{\partial u}{\partial n} v,$$  \hspace{1cm} (2.2)

Note that due to $\frac{\partial u}{\partial n} = 0$ on $\Gamma_2$ and $v = 0$ the term $\int_{\Gamma} \frac{\partial u}{\partial n} v$ disappears on $\Gamma_2$. Finally the finite method simply replaces $V_g$ with a finite element space, such as

$$V_{gh} = \{v_h \in C^0(\Omega) : \forall K \in \mathcal{T}_h, v_h|_K \in P_1 \text{ and } v_h|_{\Gamma_2} = 0\}$$

in the case of linear elements; for full detail see [10,21].

Below, we give an example of an elaborate mesh for a 3d fish and the Poisson problem; the FreeFem++ code is below and the results are on Figures 1 & 2.

```
load "msh3" load "medit" load "tetgen"

mesh3 Ths("Y5177_Fish_cut.mesh"); // read skin fish mesh ..
real hh = 10; // the final mesh size
real[3] domaine = [0,0,0,1,hh^3/6.];

mesh3 Th=tetg(Ths,switch="pqaAYY",regionlist=domaine);

fespace Xh(Th,P2);
real x0=-200,y0=0,z0=0;
func ball = sqrt((x-x0)^2+(y-y0)^2+(z-z0)^2) <30;

macro Grad(u) [dx(u),dy(u),dz(u)] // EOM

Xh p,q;
solve laplace(p,q,solver=CG) =
    int3d(Th)( ball*p*q+Grad(p)'*Grad(q) )
    - int3d(Th) ( 1*q );
plot(p,fill=1,wait=1,value=0,nbiso=50); // see figure 1

The skin mesh of the 3d fish come from the url ∗. And the graphics solution of the problem is 2d and 3d is display figure 2 and 1 respectively.

∗ http://www-roc.inria.fr/gamma/download/counter.php?dir=FISH/
&get_obj=Y5177_Fish_cut.mesh&acces=Y5177_Fish
```
3. Mesh adaptation

In FreeFem++ a lot of adaptation tools are implemented; this corresponding to the work of many people \cite{20, 26, 25, 32, 19, 29, 9, 6}.

All these tools are based a Delaunay-Voronoi algorithm with a distance (or Metric) between two points $q^1, q^2$ given by $\sqrt{t(q^1-q^2)M(q^1-q^2)}$ where the matrix $M$ is symmetric positive definite matrix field defined on the mesh $\mathcal{T}_h$. Consequently the length $\ell_{\mathcal{M}}$ of a curve $\gamma$ for $]0,1[ \subset \mathbb{R}^d$ with respect to $\mathcal{M}$ is

$$ \ell_{\mathcal{M}} = \int_0^1 \sqrt{\gamma(t)^T \mathcal{M}(\gamma(t)) \gamma(t)} dt \quad (3.1) $$

The computation of $\mathcal{M}$ can be implicit of explicit; the tools to compute $\mathcal{M}$, in isotropic cases are scalar fields $h$ which represent the local mesh size such that $\mathcal{M} = \frac{1}{h^2} I_d$, where $I_d$ is the identity matrix.

The idea is to build a meshes with edges of equal length with respect of $\mathcal{M}$, i.e. the length with respect to $\mathcal{M}$ of all the mesh edges should be closed to one so as to get an equi-repartition of the error close to $\text{err}$ in norm $L^\infty$. So for $P_1$ continuous finites elements, The metric can be defined by:

$$ \ell_{\mathcal{M}} = \frac{1}{\text{err}} |\partial^2_h u_h| \quad (3.2) $$

where $|\partial^2_h u_h| = \sqrt{(\partial^2_h u_h)^2}$ and $\partial^2_h u$ is an approximation of the Hessian matrix of $u_h$. FreeFem++ does it automatically by default with adaptmesh function (2d), MetricPk (2d), or with mshmet (2d).

An example of $L^\infty$ error mesh adaptation with metric for the Poisson problem in an L-shape domain $\Omega = ]0,1[ \setminus [1/2,1]^2$. Find $u \in H^1(\Omega)$ such that

$$ -\Delta u = (x-y) \text{ in } \Omega, \quad \frac{\partial u}{\partial n} = 0 \text{ on } \partial \Omega; \quad \int_{\Omega} u = 0 \quad (3.3) $$
we add do a small stabilization term $\varepsilon u$ to remove the constraint $\int_{\Omega} u$, so the weak form is: Find $u_\varepsilon \in H^1(\Omega)$ such that

$$\forall v \in H^1(\Omega), \quad \int_{\Omega} u_\varepsilon v + \varepsilon u_\varepsilon v - \int_{\Omega} (x - y)v = 0 \quad (3.4)$$

You can easily prove that $||u_\varepsilon - u||_{H^1} < \varepsilon C$, the discretization with $P_1$ Lagrange finite element and $\varepsilon = 10^{-10}$, is:

```cpp
int[int] lab=[1,1,1,1];
mesh Th = square(6,6,label=lab);
Th=trunc(Th,x<0.5 | y<0.5, label=1);
fespace Vh(Th,P1);
Vh u,v;
real error=0.01;
problem Poisson(u,v,sol=CG,eps=1.0e-6) =
  int2d(Th,qforder=2)( u*v*1.0e-10+ dx(u)*dx(v) + dy(u)*dy(v))
  + int2d(Th,qforder=2)( (x-y)*v);
for (int i=0;i< 4;i++)
{
  Poisson;
  plot(u,wait=1);
  Th=adaptmesh(Th,u,err=error);
  plot(Th,wait=1);
  u=u; // reinterpolation of u on new mesh Th.
  error = error/2;
}
```

The final mesh adaption and associated solution will be show in figure 3 and 4.

![Figure 3. Mesh](image3.png)  ![Figure 4. Solution](image4.png)

With the MetricPk plugin of J-M Mirebeau we can build a metric in $\mathbb{R}^2$ for the lagrangian triangular continuous approximation of degree $k$, $P_k$, with respect to the norm $W^{r,p}$ with $r = 0$ or 1; the output is an adapted mesh with $Nt$ element...
with a metric which can be optimal or sub optimal; the lost of optimality is due to constraint on acute angle in triangulation in case of \( r = 1 \). (see \[27,26\] or J-M Mirebeau these details):

```cpp
load "MetricPk"
real Nt = 1000, r = 1, k = 3, p = 1; // def of the parameter
mesh Th = square(20, 20, [2*x-1, 2*y-1]);
func f = x^2*y + y^3 + tanh(5*(-2*x + sin(5*y)));
fespace Metric(Th, [P1, P1, P1]);
Metric [m11, m12, m22];
for(int i=0; i<5; i++){
    plot(Th, wait=true); // see figure 6 and 5
    [m11, m12, m22] = [0, 0, 0]; // resize metric array
    m11[] = MetricPk(Th, f, kDeg=k, rDeg=r, pExp=p, mass=Nt / Z,
    TriangulationType=0); // 0 sub optimal
    // 1 Optimal (=>no acute angle)
    Th = adaptmesh(Th, m11, m12, m22, IsMetric=true); }
```

**Figure 5.** Optimal metric in norm \( W^{1,1} \) for 10^3 triangle but it is wrong mesh because the acute constraint is miss.

**Figure 6.** Sub Optimal but not acute constraint for 10^3 triangle.

In 3d we can use meshmet a plugin by J. Morice for the mshmet library of P. Frey \[16\]. To compute an isotropic adapted metric on can use tetgen, another plugin documented in \[30\]. The following listing is a full test of this idea.

```cpp
load "msh3" load "tetgen" load "mshmet" load "medit"
int n = 6;
int[int] ll=[1,1,1,1], l01=[0,1], l11=[1,1];
// label numbering for boundary condition
mesh3 Th3 = buildlayers(square(n, n, region=0, label=ll),
    n, zbound=[0,1], labelmid=l11, labelup = l01, labeldown = l01);
Th3 = trunc(Th3, (x<0.5) | (y < 0.5) | (z < 0.5),
```
label=1); // remove the $[0.5, 1]^3$ cube
// end of build initial mesh

fespace Vh(Th3,P1);
Vh u,v,usol;
macro Grad(u) [dx(u),dy(u),dz(u)] // EOM
solve Poisson(u,v,solver=CG) = int3d(Th3)( 1e-6*u*v + Grad(u)'*Grad(v) )
  -int3d(Th3)( (z-y/2-x/2)*v ) ;
real errm=1e-2; // level of error
for(int ii=0; ii<5; ii++) {
  Vh h ; h[]=mshmet(Th3,u,normalization=1,aniso=0,nbregul=1,
  hmin=1e-3,hmax=0.3,err=errm);
  errm*= 0.6; // change the level of error
  Th3=tetgreconstruction(Th3,switch="raAQ",sizeofvolume=h*h/6.);
  Poisson;
  plot(u,wait=1,nbiso=15); } // see figure 8

medit("U3",Th3,u); // see figure 7

Figure 7. Three 3d adapted mesh
Figure 8. iso surface

Finally, tools like splitmesh (only in 2d) can also be used: it splits triangle uniformly into sub triangles. In 3d the plugin mmg3d-v4 moves 3d meshes or build 3d anisotropic meshes inside each element (see [15]).
4. Phase change with Natural Convection

This example illustrates the coupling of natural convection modeled by the Boussinesq approximation and liquid to solid phase change in \( \Omega = [0, 1]^2 \). No slip condition for the fluid are applied at the boundary and adiabatic condition on upper and lower boundary and given temperature \( \vartheta_f \) (resp \( \vartheta_l \)) at the right and left boundaries. The starting point of the problem is Brainstorming session (part I) of the third FreeFem++ days in december, 2012, this is almost the Orange Problem is describe in web page \( \dagger \), we present the validation part with paper [35].

The main interest of this example is to show the capability of FreeFem++ to solve complex problem with different kind of non-linearity. You can find other complex example in [5, 7, 23, 18, 12, 11] for example, (Thank To. I. Danalia to the help in the modelization).

So the full model is: find \( \mathbf{u} = (u_1, u_2) \), the velocity field, \( p \) the pressure field and \( \vartheta \) the temperature field in domain \( \Omega \) such that

\[
\begin{align*}
\frac{\partial \mathbf{u}}{\partial t} + (\mathbf{u} \nabla) \mathbf{u} + \nabla p + \nabla \vartheta &= -c_T \vartheta \mathbf{e}_2 \quad &\text{in } \Omega_f \\
\nabla \cdot \mathbf{u} &= 0 \quad &\text{in } \Omega_f \\
\frac{\partial \vartheta}{\partial t} + (\mathbf{u} \nabla) \vartheta + \nabla k_T \nabla \vartheta &= \frac{\partial}{\partial t} S(T) \quad &\text{in } \Omega
\end{align*}
\]

Where \( \Omega_f \) is the fluid domain and the solid domain is \( \Omega_s = \Omega \setminus \Omega_f \). The enthalpy of the change of phase is given by the function \( S \); \( \mu \) is the relative viscosity, \( k_T \) the thermal diffusivity.

In \( \Omega_f = \{ x \in \Omega; \vartheta > \vartheta_f \} \), with \( \vartheta_m \) the melting temperature the solid has melt.

We modeled, the solid phase as a fluid with huge viscosity, so:

\[
\mu = \begin{cases} 
\vartheta < \vartheta_f \sim 10^6 \\
\vartheta \geq \vartheta_m \sim \frac{1}{Re}
\end{cases}
\]

This removes movement in the solid phase, and here Re is the Reynolds number of the liquid phase.

The Stefan enthalpy \( S_c \) with defined by \( S_c(\vartheta) = H(\vartheta) / S_{th} \) where \( S_{th} \) is the stefan number, and \( H \) is the Heaviside function ( 0, if \( \vartheta < 0 \), 1 else). with use the following smooth the enthalpy:

\[
S(\vartheta) = \frac{\tanh(50(\vartheta - \vartheta_m)))}{2S_{te}}.
\]

We apply a fixed point algorithm for the phase change part (the domain \( \Omega_f \) is fixed at each iteration) and a full no-linear Euler implicit scheme with a fixed domain for the rest. We use a Newton method to solve the non-linearity.

Remark, if we don’t make mesh adaptation the Newton method do not converge, and if we use explicit method this diverge too, and finally if we implicit the dependance in $\Omega_s$ the method also diverge. This is a really difficult problem.

The finite element space to approximate $u_1, u_2, p, \vartheta$ is defined by

```csharp
fespace Wh(Th, [P2,P2,P1,P1]);
```

We do mesh adaptation a each time step, with the following code:

```csharp
Ph ph = S(T), pph=S(Tp);
Th= adaptmesh(Th,T,Tp,ph,pph,[u1,u2],err=errh,
 hmax=hmax, hmin=hmax/100,ratio = 1.2);
```

This mean, we adapt with all variable plus the 2 melting phase a time $n+1$ and $n$ and we smooth the metric with a ratio of 1.2 to account for the movement of the melting front.

The Newton loop and the fixed point are implemented as follows

```csharp
real err=1e100,errp;
for(int kk=0;kk<2;++kk) // 2 step of fixe point on $\Omega$
{ nu = nuT; // recompute the viscosity in $\Omega_s, \Omega_f$
 for(int niter=0;niter<20; ++niter) // newton loop
 { BoussinesqNL;
 err = u1w[].linfty;
 cout << niter << "err_NL" << err <<endl;
 u1[] -= u1w[];
 if(err < tolNewton) break; }// convergence ..
}
```

The linearized problem is

```csharp
problem BoussinesqNL([u1w,u2w,pw,Tw],[v1,v2,q,TT]) =
int2d(Th) ( [u1w,u2w,Tw]'*[v1,v2,TT]*cdt
 + UgradV(u1,u2,u1w,u2w,Tw)' * [v1,v2,TT]
 + UgradV(u1w,u2w,u1,u2,T)' * [v1,v2,TT]
 + ( Grad(u1w,u2w)'*Grad(v1,v2) ) * nu
 + ( Grad(u1,u2)'*Grad(v1,v2)) * dnu* Tw
 + cmT*Tvv2
 + grad(Tw)'*grad(TT)*kT
 - div(u1w,u2w)*q -div(v1,v2)*pw - eps*pw*q
 + dS(T)*Tw*TT*cdt
 )
 - int2d(Th) ( [u1,u2,T]'*[v1,v2,TT]*cdt
 + UgradV(u1,u2,u1,u2,T)' * [v1,v2,TT]
 + ( Grad(u1,u2)'*Grad(v1,v2)) * nu
 + cmT*T*v2
 + grad(T)'*grad(TT)*kT
 - div(u1,u2)*q -div(v1,v2)*p
 - eps*p*q // stabilisation term
```
\[ + S(T) \cdot TT \cdot cdt \]
\[ - [u1p, u2p, Tp'] \cdot [v1, v2, TT] \cdot cdt \]
\[ - S(Tp) \cdot cdt \cdot TT \]
\)
\[ + \text{on}(1, 2, 3, 4, u1w=0, u2w=0) + \text{on}(2, Tw=0) + \text{on}(4, Tw=0) \];

The parameters of the computation are taken from [35] case 2. \( \theta_m = 0, Re = 1, \]
\( S_{te} = 0.045, P_f = 56.2, R_a = 3.2710^5, \theta_l = 1, \theta_r = -0.1 \) so in this case cmT = \( c_T = -R_a / P_f \), \( kT = kT = 1 / P_f \), eps = \( 10^{-6} \), time step \( \delta t = 10^{-1} \), cdt = \( 1 / \delta t \), and with a good agreement this is figure 6 of [35] at time \( t = 80 \); see in figures 9 and 10.

![Figure 9. Iso-value of the temperature at time \( t = 80 \) of problem (4.1)](image)

![Figure 10. Velocity at time 80 of problem (4.1)](image)

### 5. A Schwarz Domain Decomposition Example in Parallel

The following is an explanation of the scripts DDM-Schwarz-* . edp of the distribution. This is Schwarz domain decomposition in parallel with a complexity almost independent of the number of process (with a coarse grid preconditioner), thanks to F. Nataf for the all the discussion to implementation of this algorithm (see [28] for the theory).

To solve the following Poisson problem

on domain \( \Omega \) with boundary \( \Gamma \) in \( L^2(\Omega) \):

\[-\Delta u = f, \text{ in } \Omega, \text{ and } u = g \text{ on } \Gamma,\]

where \( f \) and \( g \) are two given functions of \( L^2(\Omega) \) and of \( H^1(\Gamma) \),
Let \((\pi_i)_{i=1,\ldots,N_p}\) be a regular partition of unity of \(\Omega\), i.e.:

\[
\pi_i \in C^0(\Omega) : \quad \pi_i \geq 0 \quad \text{and} \quad \sum_{i=1}^{N_p} \pi_i = 1.
\]

Denote \(\Omega_i\) the sub domain which is the support of \(\pi_i\) function and also denote \(\Gamma_i\) the boundary of \(\Omega_i\). The parallel Schwarz method is as follows.

Let \(\ell = 0\) be the iterator and \(u^0\) an initial guest respecting the boundary condition (i.e. \(u^0|_{\Gamma} = g\)):

\[
\forall i = 1,\ldots,N_p : \quad -\Delta u^\ell_i = f, \quad \text{in} \quad \Omega_i, \quad \text{and} \quad u^\ell_i = u^\ell \quad \text{on} \quad \Gamma_i \setminus \Gamma, \quad u^\ell_i = g \quad \text{on} \quad \Gamma_i \cap \Gamma \quad (5.1)
\]

\[
u^\ell+1 = \sum_{i=1}^{N_p} \pi_i u^\ell_i \quad (5.2)
\]

After discretization with the Lagrange finite element method, with a compatible mesh \(T_{hi}\) for \(\Omega_i\), there exists a global mesh \(T_h\) such that \(T_{hi}\) is included in \(T_h\). Remark We avoid using finite element spaces associated to the full domain \(\Omega\) because it is too expensive.

Now let us look at this academic example:

\[-\Delta u = 1, \quad \text{in} \quad \Omega = [0,1]^3; \quad u = 0, \quad \text{on} \quad \partial \Omega. \quad (5.3)\]

In the following test we use a GMRES version preconditioned by the Schwarz algorithm \((P_3)\) and with a coarse grid solver \((P_c)\) on a coarse mesh. To build a new preconditioner \(P\) from two preconditioner \(P_c\) and \(P_s\) we use the following approximation \(0 \sim I - AP^{-1} = (I - AP_c^{-1})(I - AP_s^{-1})\) and after a simple calculus we have \(P^{-1} = P_c^{-1} + P_s^{-1} - P_c^{-1}AP_s^{-1}\).

6. Conclusion

FreeFem++ is a continuously evolving software because it is easy to add new tools, finite elements. We have try to illustrate this with three example. In the future we expect to include more tools for try dimensional mesh anisotropic meshes adaptive, automatic different ion by operator overloading, seamless integration of parallel tools to free the user off low level programming.

Acknowledgements Thanks to O. Pironneau for the fruitful discussion and the initialization for the freefem storye, to K. Ohtsuka to the help the writing of the documentation, to A Le Hyaric for the integrated version FreeFem++-cs, to J. Morice, and S. Auliac for the development respectively of the three dimensional meshes, and the optimization toolbox, and finaly to E. Chacòn Vera for the freefem++ wiki.

\[http://www.um.es/freefem/ff++/pmwiki.php\]
Figure 11. Resolution of Poisson equation with 1 728 000 DoF, scalability curve

Figure 12. On 48 proc, test on number of DoF from $216 \times 10^3$ to $216 \times 10^6$, complexity curve
References


